SolidWorks macro step by step

Level: Beginner to intermediate

Macros are a great way to enhance the functionality of SolidWorks. In this article we will try to dig some of the major stumbling blocks when you begin macro coding in SolidWorks.

I assume that you are familiar with the programming of basic VBA macros and have understood the meaning of declarations, object model, procedures, functions and other stuff like this. If you are not, please take a look at other "How To-articles" which addresses these issues or look for standard documentation or training lessons regarding VB and VBA programming.

No for the funny part of this page. In this example you will learn how to ...

· plan your macro and dig for the starting points

· traverse all views of a sheet

· get the models of a view

· switch configurations of models

· open documents in SolidWorks

· get mass properties

· work with safe arrays

· work with your own functions

Step 1: Plan your macro

This is important, really. Like you should have a clear design intend when modeling parts or assemblies, you should know what you want your macro to do. First, before ever looking at API help or record a macro make sure that you know what you are looking for! Make a brief description of what you want your macro to do. As you will see commenting the code is not only a good practice but will help you and others to understand your code and will force you to think about what you have done. So lets make a description of what we want to achieve:

Macro gets the mass for the model in the view of a drawing

Well, lets think about that a little bit as a programmer. You are now on the other side of the edge, and you have to imagine what a user would want to do and how he/she would use your macro. So what drawing would the typical user assume? Either he should be prompted for what drawing he wants to check or the drawing should be already active. For it is much easier we choose the second way ;)

You know that there could be several sheets in a drawing and several (or no) views in each of this sheets. Each view can hold a different model; or there may be some views which holds the same model, but different configurations of it. You have to make sure that you check for all of this if you want to build a robust macro. For the purpose of this example we want to access all models in all views of the active sheet, regardless of its configuration. With this in mind lets be more specific:

Macro gets the mass for the model configuration in all views of the active sheet in the active drawing 

Now lets begin with the macro, we need a starting point. Like described in other How-To-articles either record a macro, stop it and reopen it for edit, copy any other macro to a new name or just hit Tools/Macro/New if you already use SolidWorks 2003. I'm not a friend of unnecessary references and also don't want to blow up my macros file with the complete constants of swconst.bas, so there are no references to the SolidWorks type libraries (and therefor all declarations "as Object") nor an imported swconst.bas If you think it is better practice the other way round feel free to do so, it wont have a blow back, if you are doing this only for yourself.

Step 2: Riding the horse backwards

In most situations it is easier to dig the API calls from bottom to top to achieve our intended goal. We have to look for a call which provides us with the mass properties. Remember: the API help file is your best friend. I can't stress that enough: it's good practice to analyze  code written by others or ask for help and look some websites (like www.solidworksdeveloper.com ), but the API help file is your best friend (and sometimes the only one). So open the help file and search for "mass properties" and you will see an example in VB; wow. So there is out entry point, we have to use ModelDoc.GetMassProperties or its successors to obtain the mass. 

But what means, that the properties are returned as "a VARIANT of type Safearray"? Don't get lost, this is very simple, the return value is an array packed inside a variant. Just declare your return value "as Variant" and use the result as an array. The description says, the result will be a array of 12 doubles, zero-based as usual, so we can access them with MassProp(0) to MassProp(11).

A macro to this point would look like this:

Dim ModelDoc As Object

Dim MassProp As Variant

Dim Mass As Double

Sub main()

      ' Now we can get the mass properties. The function will return

      ' a array of 12 doubles packed as safearray ... what the heck ...

      ' It's that easy: declare as Variant, use as array of doubles :-))

      MassProp = ModelDoc.GetMassProperties

      Mass = MassProp(5)

End Sub

Well, wont execute, but looks nice.

But how do we get the ModelDoc object? We must attach to the model, which is referenced in the view we want to examine, but how to get it? Ask your friend: open API help, search for "referenced model in view" and browse the answers. You are presented a call called View.GetReferencedModelName which returns the name of the model. In fact, it is the complete path to the model including the filename extension. But it's only the name, not the ModelDoc object, so we have to achieve the object for that model. A quick look at the API help file for "get modeldoc object by name" reveals several OpenDoc... calls. For this article I choose the easiest (and most obsolete one) SldWorks.OpenDoc(Name, Type), in the code there is also the most recent one OpenDoc6.

OpenDoc will work fast, for usually the models are loaded into memory by SolidWorks when you opens a drawing. With OpenDoc you have to know the name of the file (and we have gathered that already) and the type of file. That's difficult, there are no tricks (at least I don't know any) to find out the type of a file. There is only one solution: we have to rely on the filenames extension. For we can use this each time we use OpenDoc lets build a VBA function:

Const swDocNONE = 0      '  Used to be TYPE_NONE

Const swDocPart = 1       '  Used to be TYPE_PART

Const swDocASSEMBLY = 2      '  Used to be TYPE_ASSEMBLY

Const swDocDRAWING = 3       '  Used to be TYPE_DRAWING

Const swDocSDM = 4       '  Solid data manager.

Function SWXTypeOfFile(filename As String) As Long

    ' function will determine the filetype used by OpenDoc

    ' by analysing the filename extension; the SolidWorks

    ' constants have to be imported with swconst.bas or declared

    Select Case UCase(Right(filename, 6))

    Case "SLDPRT"

        SWXTypeOfFile = swDocPart

    Case "SLDASM"

        SWXTypeOfFile = swDocASSEMBLY

    Case "SLDDRW"

        SWXTypeOfFile = swDocDRAWING

    Case Else

        SWXTypeOfFile = swDocNONE

    End Select

End Function
A big ATTENTION here: prior to SolidWorks 2003 the starting procedure function (usually Sub main) MUST (MUST!!!!) be the last procedure in the last included module (= not the last one in project, they are sorted in alphabetic order, but the last module you included) to work properly when you run the macro directly from Tools/Macros/Run. Therefor all other functions must be included before Sub main

...

Dim swApp As Object

Dim View As Object

Dim RefModelName As String

...

Sub main()

   ...

   RefModelName = View.GetReferencedModelName

   Set ModelDoc = swApp.OpenDoc(RefModelName, SWXTypeOfFile(RefModelName))

   ...

End Sub
Now for the configuration: the mass properties depends on the configuration of the model, so we have to switch the active configuration of the model to the one referenced in the view. Let's ask our friend again: "switch configuration" brings up the call ModelDoc2.ShowConfiguration2(configurationName) and "referenced configuration in view" dig out  View.ReferencedConfiguration. So everything at hand, let's summarize so far:

RefModelName = View.GetReferencedModelName

ConfigName = View.ReferencedConfiguration

Set ModelDoc = swApp.OpenDoc(RefModelName, SWXTypeOfFile(RefModelName))

errors = ModelDoc.ShowConfiguration2(ConfigName)

MassProp = ModelDoc.GetMassProperties

Last thing to do: we want to traverse all views of the active sheet. Searching the API help for "drawing views current sheet" reveals DrawingDoc.GetFirstView and View.GetNextView. Don't be surprised to see a DrawingDoc object, this is only a special derived ModelDoc-object which only contains methods and properties solely for drawings.

The task to build traversal loop for objects is common, so the shown method can easily be adapted to other traversals. The trick is not to test if there is a next object, but just to attach to it and check if you get an object or not. If SolidWorks can't return an object the object variable is set to the special value Nothing, which can be looked for:

Sub main()

    Set swApp = CreateObject("SldWorks.Application")

    Set DrawingDoc = swApp.ActiveDoc

    Set View = DrawingDoc.GetFirstView

    ' as long as there is a valid view

    While Not View Is Nothing

        ' now do something useful with the view object

        ...


     ' get the next drawing view

        Set View = View.GetNextView

    Wend

End Sub
Step 3: Running and testing

Now that we have all parts together we can build the macro and begin testing. A tip: SolidWorks VBA editor environment automatically saves all changes to the code, so you wont need the save icon. But this is also true if you just exit the editor! If you have messed up your code you have to repair it, don't close the editor and hope to get your old working code back, for you haven't saved it.

If you want to test something, don't delete or change it but copy the code and uncomment the parts of the code you want to get rid of.  If you want to see your plain code export it to *.bas or *.frm files; theses are plain ASCII and can be opened in any editor. And if you have working code and want to enhance it, make a backup copy of your macro prior to work.

Then test your code. Try to think as a user who don't know your intend and try to break your macro with normal means. If you find an error and think you as a user wont accept it fix it. That's the easy idea  behind good programming style. You should have commented your code on the fly, if you missed it, do it now. It will help. Not only others, but it will help yourself to understand what you have done and why you have it done this way.

You can download the complete commented macro either from here or from my website dedicated to tools and macros http://swtools.cad.de . Look for the macro section and download the one labeled mm_21.zip . 

I hope I don't wasted your time and you get some of my ideas.

Stefan 

